
Schrodinger's Equation

One Dimension

This is our twenty-third notebook. The goal for the remainder of the course will be to use Mathematica to help us solve and visualize the solutions to Schrodinger's equation for a few different systems.


Let's start with our old friend, the one-dimensional harmonic oscillator.

Potentials

Schrodinger's equation is not written in terms of forces. It is written in terms of potentials. For a spring with spring constant k , the potential is:

```
In[4]:= springPotential[x_] := Module[{k = 1},  $\frac{1}{2} k x^2$ ]
```

```
Plot[springPotential[x], {x, -2, 2}]
```



You might completely reasonably ask what on Earth this potential has to do with the force

$F(x) = -kx$. The deep relationship between potentials and forces is that the force is minus the derivative of the potential. Let's have Mathematica compute the minus the derivative for us:

```
In[6]:= potential[x_] :=  $\frac{1}{2} k x^2$ 
```

```
-Derivative[1][potential][x]
```

```
Out[7]= -k x
```

Ok, the relationship is as claimed. You are just going to have to accept that Schrodinger used potentials as a way of introducing forces in quantum mechanics. Before Schrodinger, de Broglie and Bohr had made some interesting, crude steps, but neither had written down an equation that could work for real systems like the harmonic oscillator and the hydrogen atom.

If you have a couple of semesters of classical mechanics (one is not quite enough), what Schrodinger did seems entirely reasonable, and even after one semester of classical mechanics it begins to seem reasonable, but it was still a stunning intellectual leap for Schrodinger to combine potentials and second derivatives when writing down the quantum-mechanical equation for the electron.

Schrodinger's Equation

It is common to use the letter ψ for the wave function. Here is the differential equation that ψ satisfies:

```
In[8]:= -  $\frac{\hbar^2}{2m}$  Derivative[2][ $\psi$ ][x] +  $\frac{1}{2} k x^2[x] \psi[x] == \text{energy} \psi[x]$  // TraditionalForm
```

$$-\frac{\hbar^2}{2m} \frac{d^2 \psi(x)}{dx^2} + \frac{1}{2} k x^2 \psi(x) = \text{energy} \psi(x)$$

```
Out[8]//TraditionalForm=
```

Comparison with Wave Equations

Let's simplify for a moment. The $\frac{1}{2} k x^2$ is unlike anything we have had in a wave equation before. So let's just set $k = 0$ for a moment and look at what remains:

```
In[9]:= -  $\frac{\hbar^2}{2m}$  Derivative[2][ $\psi$ ][x] == \text{energy} \psi[x] // TraditionalForm
```

$$-\frac{\hbar^2}{2m} \frac{d^2 \psi(x)}{dx^2} = \text{energy} \psi(x)$$

```
Out[9]//TraditionalForm=
```

If energy < 0, this boils down to:

$$\psi''(x) \propto \psi(x)$$

That kind of equation is solved by rising or falling exponentials. If energy > 0, this boils down to:

$$\psi''(x) \propto -\psi(x)$$

That kind of equation is solved by sines and cosines.

Reintroducing the potential term, which near the center of the harmonic oscillator is small and in far away regions is large, causes the Schrodinger equation solutions to look like waves near the center of the harmonic oscillator, and like exponentially-damped (or exponentially-growing) functions far way from the center.

There is an entire semi-rigorous analysis based on the idea in the preceding paragraph, known as the WKB approximation. We are not going to have to study the WKB approximation, because we have Mathematica's `NDSolve[]` to do the hard work of solving an otherwise challenging differential equation for us.

Units

I am about to set the very important constants (“hbar”, the particle mass, and the spring constant) all equal to one. You might ask why I am not about to set energy = 1 as well. The reason is that that over-constrains the problem.

Imagine setting both the radius of a sphere equal to one, and the circumference of that same sphere equal to one. You’d quickly get garbage in whatever problem in which you did that. You can set the radius to one, in which case the circumference is 2π , or you can set the circumference to one, in which case the radius is $\frac{1}{2\pi}$, but you can’t set them both to one. What we are doing when we set a dimensionful quantity like the radius to one is actually choosing our units of length. We cannot then set some other unrelated or non-trivially related length to one.

When we set $\hbar = k = m = 1$, we are choosing our units of angular momentum, force, and mass. We are then unable to also set energy = 1, because the combination $\hbar \sqrt{\frac{k}{m}}$ has units of energy.

Note that once again the combination $\sqrt{\frac{k}{m}}$ has shown up, and just as we did when we first encountered this combination in the classical harmonic oscillator, we can give it the name ω , in which case our units of energy when we set all these constants to one are $\hbar\omega$.

Boundary Conditions

Very strangely, in quantum mechanics particles can be where they have no business being classically. If a particle has energy E , then it has no business being further from the origin than the two solutions of $E = \frac{1}{2} kx_{\pm}^2$. Beyond x_{\pm} , the particle is forbidden. If this were also true in quantum mechanics, our boundary conditions would be to just set $\psi(x_+) = \psi(x_-) = 0$.

In quantum mechanics, unless the potential gets infinitely high somewhere, the particle has some very small but nonzero chance of being far into the region it has no business being classically.

Let’s go a long ways (3 in our units will be enough) into the forbidden region and demand that $\psi(x)$ be zero there. Our left boundary condition will be:

```
In[10]:= longWays = 3;
psi[-longWays] == 0

Out[11]= psi[-3] == 0
```

You know with equations with second derivatives you have to specify more than just one edge’s conditions. With the guitar string, we had to specify a boundary condition at the bridge and the nut. With the drumhead, we had to specify a boundary condition all around the edge. With this problem (for reasons

you will soon see, but accept for the moment), the second boundary condition will be:

```
In[12]:= Derivative[1][psi][-longWays] == 0.1
Out[12]= psi'[-3] == 0.1
```

If your gut is saying this is arbitrary, your gut is right. Just accept it for the moment.

The Full Problem

Ok, we have the full problem, including boundary conditions now:

```
In[13]:= harmonicOscillatorProblem = Module[{ħ = 1, m = 1, k = 1},
  { - (ħ^2 / (2 m)) Derivative[2][psi][x] + (1/2) k x^2 psi[x] == energy psi[x],
    psi[-longWays] == 0, Derivative[1][psi][-longWays] == 0.1 }];
harmonicOscillatorProblem // TraditionalForm
Out[14]//TraditionalForm=
```

$$\left\{ -\frac{\hbar^2}{2m} \psi''(x) + \frac{1}{2} k x^2 \psi(x) = \text{energy} \psi(x), \psi(-3) = 0, \psi'(-3) = 0.1 \right\}$$

Note that time has not entered. This is another thing for me to discuss/explain, but because it hasn't entered, you shouldn't be bothered that we don't have any initial conditions.

Making Mathematica Solve the Problem


I am going to stick the entire problem into a Manipulate[] so that we can play with the one remaining variable, which is the energy:

```

In[15]:= Manipulate[
  Module[{psiSolution = NDSolveValue[harmonicOscillatorProblem /. energy -> e,
    psi, {x, -longWays, longWays + 1}]},
    Plot[psiSolution[x], {x, -longWays, longWays + 1},
      PlotRange -> {{-longWays, longWays + 1}, {-2, 2}}, PlotLabel -> e]], {e, 0, 5}]

```

Out[15]=



Boundary Conditions Revisited

Boundary conditions are handled strangely in the Schrodinger's equation. I mentioned that

```

In[16]:= Derivative[1][psi][-longWays] == 0.1 // TraditionalForm

```

Out[16]//TraditionalForm=
 $\psi'(-3) = 0.1$

was arbitrary. The actual boundary condition we are looking for to complement $\psi[-\text{longWays}] == 0$ is $\psi[+\text{longWays}] == 0$. It turns out it is hard to make both of those things true. What you need to do is fiddle with the energy until $\psi(+\text{longWays})$ is as close to 0 as you can make it.

Record the Energy Levels

I chose $\text{longWays}=3$ because a larger longWays would require more precision than you can come close to achieving with the `Manipulate[]` control.


Fiddle with the control in the `Manipulate[]`. You should be able to find five energy levels for which $\psi[+\text{longWays}]$ is pretty close to 0. Record them below, in ascending order:

"Level (n)"	"Energy (units of $\hbar\omega$)"
0	<input type="checkbox"/>
1	<input type="checkbox"/>
2	<input type="checkbox"/>
3	<input type="checkbox"/>
4	<input type="checkbox"/>

It would be very nice to make crude sketches of the functions as you use Mathematica to find them.

Note that we have labeled the solution we have found with an integer n , starting with $n = 0$. We could go beyond $n = 4$, but 4 is high enough to see the patterns emerge, and I would have had to make longWays bigger to get decently accurate results for larger values of n . Do not be concerned about the vertical axis in your sketches. We'll interpret the vertical axis in the next notebook.

```
In[18]:= Table[Plot[{}, {x, -3, 3}], 5] // Column
Out[18]=
```



Where Did Time Go?

In the real world, even the quantum-mechanical real world, things evolve with time. If we have time, I can roll up the screen and explain, using the separation of variables trick, where time disappeared to.

The equation that I have been calling “Schrodinger’s Equation” is, more precisely, known as the “Time-Independent Schrodinger Equation.” It is applicable to a system that has a definite energy, and solving it allows you to discover the allowable energies of the system. A general system, can have a mixture of energies. However, the mixture still only contains mixtures of the allowed energies.

If we are short of time, we can just proceed to the next notebook, where I will show you how to interpret the solutions $\psi_n(x)$ and how combinations of solutions are added together to recover time-dependence.